

Examining Lagged Effects of Stress in Predicting Psoriasis Severity

These data come from a study of weekly fluctuation in psoriasis severity, as also featured in Example 4. The goal of these analyses is to examine how weekly fluctuations in psoriasis severity may be predicted by weekly fluctuation in stress, and at what time lag these effects might be observed.

SAS Data Setup for Lags:

```
* Location for original SPSS portable multivariate file for these models;
LIBNAME import SPSS "F:\Example Data\p2mult.por";
DATA work.p2stack; SET import.p2mult;
  * Calculating person mean for time-varying predictor;
  pmStress = MEAN(of lesneg1-lesneg8); LABEL pmStress = "Person-Mean Stress";
  * Stacking data;
  Week=1; severity=sapasi1; stress=lesneg1; OUTPUT;
  Week=2; severity=sapasi2; stress=lesneg2; OUTPUT;
  Week=3; severity=sapasi3; stress=lesneg3; OUTPUT;
  Week=4; severity=sapasi4; stress=lesneg4; OUTPUT;
  Week=5; severity=sapasi5; stress=lesneg5; OUTPUT;
  Week=6; severity=sapasi6; stress=lesneg6; OUTPUT;
  Week=7; severity=sapasi7; stress=lesneg7; OUTPUT;
  Week=8; severity=sapasi8; stress=lesneg8; OUTPUT;
  LABEL week = "Week of Study" severity = "Psoriasis Severity"
  stress = "Weekly Stress"; RUN; %LET datafile=p2stack;

* Creating predictors for analysis;
DATA p2stack; SET p2stack; WHERE NMISS(severity, stress)=0; * Selecting complete cases;
* Creating person mean(between-person) indicator for LES;
  PMstress2 = pmStress - 2; LABEL PMstress2 = "Between-Person Stress (0=2)";
* Person-centering stress to be a level 1 predictor;
  WPstress = stress - pmStress; LABEL WPstress = "Within-Person Stress (0=person mean)";
* Nothing new needed for lag=0;
  WPstressLag0 = WPstress; LABEL WPstressLag0= "Within-Person Stress Lag=0";
* New values for each lag get transferred only if the row is for the same subid;
  IDlag1 = lag1(subid); WPstressLag1 = lag1(WPstress);
  IF subid=IDlag1 THEN WPstressLag1 = WPstressLag1; ELSE WPstressLag1= .;
  LABEL WPstresslag1= "Within-Person Stress Lag=1";
  IDlag2 = lag2(subid); WPstressLag2 = lag2(WPstress);
  IF subid=IDlag2 THEN WPstressLag2 = WPstressLag2; ELSE WPstressLag2= .;
  LABEL WPstressLag2= "Within-Person Stress Lag=2"; run;
```

Example of Data File after Lagging:

subid	Week	severity	pmStress	PMstress2	stress	WPstress	WPstressLag0	IDlag1	WPstressLag1	IDlag2	WPstressLag2
100	1	1.93	0.60	-1.40	0.00	-0.60	-0.60
100	2	2.00	0.60	-1.40	0.00	-0.60	-0.60	100	-0.60	.	.
100	3	1.85	0.60	-1.40	1.10	0.50	0.50	100	-0.60	100	-0.60
100	4	1.68	0.60	-1.40	0.69	0.09	0.09	100	0.50	100	-0.60
100	5	1.82	0.60	-1.40	0.00	-0.60	-0.60	100	0.09	100	0.50
100	6	0.00	0.60	-1.40	1.39	0.79	0.79	100	-0.60	100	0.09
100	7	1.47	0.60	-1.40	1.61	1.01	1.01	100	0.79	100	-0.60
100	8	0.66	0.60	-1.40	0.00	-0.60	-0.60	100	1.01	100	0.79
101	1	2.10	0.45	-1.55	0.69	0.24	0.24	100	.	100	.
101	2	1.70	0.45	-1.55	1.10	0.64	0.64	101	0.24	100	.
101	3	1.50	0.45	-1.55	0.00	-0.45	-0.45	101	0.64	101	0.24
101	4	2.00	0.45	-1.55	0.00	-0.45	-0.45	101	-0.45	101	0.64
101	5	1.42	0.45	-1.55	0.00	-0.45	-0.45	101	-0.45	101	-0.45
101	6	1.26	0.45	-1.55	0.00	-0.45	-0.45	101	-0.45	101	-0.45
101	7	0.98	0.45	-1.55	1.39	0.93	0.93	101	-0.45	101	-0.45
101	8	.	0.45	-1.55	.	.	.	101	0.93	101	-0.45

1a) Empty Means, Random Intercept Model for Severity (DV)

```
TITLE 'Empty Means, Random Intercept Model for Severity Outcome';
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 METHOD=REML;
  CLASS subid week;
  MODEL severity = / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / VCORR TYPE=UN SUBJECT=subid;
  REPEATED week / TYPE=VC SUBJECT=subid; RUN;
```

$$\begin{aligned} \text{Level 1: Severity}_{ti} &= \beta_{0i} + e_i \\ \text{Level 2: } \beta_{0i} &= \gamma_{00} + U_{0i} \end{aligned}$$

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	SUBID	0.6667	0.08782	7.59	<.0001
week	SUBID	0.1318	0.006780	19.44	<.0001

ICC for Severity:
 $.6667 / (.6667 + .1318) = .83$
 Houston, we may have a problem

1b) Empty Model for Stress (Time-Varying Predictor)

```
TITLE 'Empty Means, Random Intercept Model for Stress TV Predictor';
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 METHOD=REML;
  CLASS subid week;
  MODEL stress = / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / VCORR TYPE=UN SUBJECT=subid;
  REPEATED week / TYPE=VC SUBJECT=subid; RUN;
```

$$\begin{aligned} \text{Level 1: Stress}_{ti} &= \beta_{0i} + e_i \\ \text{Level 2: } \beta_{0i} &= \gamma_{00} + U_{0i} \end{aligned}$$

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	SUBID	0.3562	0.05102	6.98	<.0001
week	SUBID	0.2519	0.01298	19.41	<.0001

ICC for Stress:
 $.3562 / (.3562 + .2519) = .56$
 At least stress is time-varying!

Our best-fitting unconditional model for time (as found in Example 4) included a random intercept variance in the **G** matrix and a lag-3 correlation with heterogeneous residual variances in the **R** matrix. We now add fixed effects of predictors to that model.

2) Predicting Severity from Between-Person Effect of Stress

$$\begin{aligned} \text{Level 1: Severity}_{ti} &= \beta_{0i} + e_{ti} \\ \text{Level 2: } \beta_{0i} &= \gamma_{00} + \gamma_{01} (\overline{\text{Stress}}_i - 2) + U_{0i} \end{aligned}$$

```
TITLE "SAS: BP Effect of Stress";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 METHOD=REML;
  CLASS subid week;
  MODEL severity = PMstress2 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=subid;
  REPEATED week / TYPE=TOEPH(4) SUBJECT=subid; RUN;
```

Dimensions

Subjects	124
Max Obs Per Subject	8

Number of Observations

Number of Observations Read	880
Number of Observations Used	880
Number of Observations Not Used	0

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	1.9464	0.1272	123	15.31	<.0001
PMstress2	0.4572	0.1097	123	4.17	<.0001

Interpret each fixed effect:

3a) Predicting Severity from Between- and Within-Person (Simultaneous) Effects of Stress

Level 1: $Severity_{ti} = \beta_{0i} + \beta_{1i} (\text{Lag0Stress}_{ti} - \overline{\text{Stress}}_i) + e_{ti}$
 Level 2: Intercept: $\beta_{0i} = \gamma_{00} + \gamma_{01} (\overline{\text{Stress}}_i - 2) + U_{0i}$
 Lag-0 WP Stress: $\beta_{1i} = \gamma_{10}$

```
TITLE "Add WP Effect of Stress at 0-Week Lag";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 METHOD=REML;
  CLASS subid week;
  MODEL severity = PMstress2 WPstressLag0 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=subid;
  REPEATED week / TYPE=TOEPH(4) SUBJECT=subid; RUN;
```

Dimensions		
Subjects	124	Note: 8 observations per person are used here.
Max Obs Per Subject	8	

Number of Observations	
Number of Observations Used	880
Number of Observations Not Used	0

Solution for Fixed Effects					
		Standard			
Effect	Estimate	Error	DF	t Value	Pr > t
Intercept	1.9463	0.1272	123	15.30	<.0001
PMstress2	0.4572	0.1097	123	4.17	<.0001
WPstressLag0	-0.00720	0.02348	680	-0.31	0.7591

Interpret the effect of WPstressLag0:

3b) Predicting Severity from Between- and Within-Person (+1-Week Lag) Effects of Stress

Level 1: $Severity_{ti} = \beta_{0i} + \beta_{1i} (\text{Lag0Stress}_{ti} - \overline{\text{Stress}}_i) + \beta_{2i} (\text{Lag1Stress}_{ti} - \overline{\text{Stress}}_i) + e_{ti}$
 Level 2: Intercept: $\beta_{0i} = \gamma_{00} + \gamma_{01} (\overline{\text{Stress}}_i - 2) + U_{0i}$
 Lag-0 WP Stress: $\beta_{1i} = \gamma_{10}$
 Lag-1 WP Stress: $\beta_{2i} = \gamma_{20}$

```
TITLE "Add WP Effect of Stress at 1-Week Lag";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 METHOD=REML;
  CLASS subid week;
  MODEL severity = PMstress2 WPstressLag0 WPstressLag1
    / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=subid;
  REPEATED week / TYPE=TOEPH(4) SUBJECT=subid; RUN;
```


Dimensions		
Subjects	122	Note the change in number of observations used: only 7 occasions per person in adding a 1-week lag within-person predictor. Estimating models on different samples means we cannot compare variance components (e.g., compute pseudo-R ²) or do any kind of LRT for change in model fit.
Max Obs Per Subject	7	
Number of Observations		
Number of Observations Used	756	
Number of Observations Not Used	124	

Solution for Fixed Effects
Standard

Effect	Estimate	Error	DF	t Value	Pr > t
Intercept	1.9649	0.1307	121	15.04	<.0001
PMstress2	0.4833	0.1132	121	4.27	<.0001
WPstressLag0	0.01976	0.02662	557	0.74	0.4582
WPstressLag1	0.05919	0.02509	545	2.36	0.0187

Interpret the effect of WPstressLag1:

Calculated predicted values in excel worksheet to plot these effects:



3b) Predicting Severity from Between- and Within-Person (+2-Week Lag) Effects of Stress

Level 1: $Severity_{ti} = \beta_{0i} + \beta_{1i} (Lag0Stress_{ti} - \overline{Stress}_i) + \beta_{2i} (Lag1Stress_{ti} - \overline{Stress}_i) + \beta_{3i} (Lag2Stress_{ti} - \overline{Stress}_i) + e_{ti}$

Level 2: Intercept: $\beta_{0i} = \gamma_{00} + \gamma_{01} (\overline{Stress}_i - 2) + U_{0i}$

Lag-0 WP Stress: $\beta_{1i} = \gamma_{10}$

Lag-1 WP Stress: $\beta_{2i} = \gamma_{20}$

Lag-2 WP Stress: $\beta_{3i} = \gamma_{30}$

```
TITLE "Add WP Effect of Stress at 2-Week Lag";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 METHOD=REML;
  CLASS subid week;
  MODEL severity = PMstress2 WPstressLag0 WPstressLag1 WPstressLag2
    / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=subid;
  REPEATED week / TYPE=TOEPH(4) SUBJECT=subid; RUN;
```

Dimensions	
Subjects	119
Max Obs Per Subject	6
Number of Observations	
Number of Observations Used	634
Number of Observations Not Used	246

Note the change in number of observations used: only 6 occasions per person in adding a 2-week lag within-person predictor.

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	1.9295	0.1397	116	13.81	<.0001
PMstress2	0.4771	0.1199	117	3.98	0.0001
WPstressLag0	0.01772	0.03117	475	0.57	0.5699
WPstressLag1	0.08317	0.02946	380	2.82	0.0050
WPstressLag2	0.04163	0.03089	489	1.35	0.1785

4) Examining Fixed Effects under Alternative Models for the Variances (Listed in Order of Fit)

<p>Unstructured R:</p> <p>Fit Statistics</p> <p>-2 Res Log Likelihood 875.6</p> <p>AIC (smaller is better) 931.6</p> <p>AICC (smaller is better) 933.8</p> <p>BIC (smaller is better) 010.1</p>	<p>Solution for Fixed Effects</p> <p>Standard</p> <table border="0"> <tr> <th>Effect</th> <th>Estimate</th> <th>Error</th> <th>DF</th> <th>t Value</th> <th>Pr > t </th> </tr> <tr> <td>Intercept</td> <td>1.9860</td> <td>0.1274</td> <td>120</td> <td>15.59</td> <td><.0001</td> </tr> <tr> <td>PMstress2</td> <td>0.4723</td> <td>0.1107</td> <td>122</td> <td>4.27</td> <td><.0001</td> </tr> <tr> <td>WPstressLag0</td> <td>0.02335</td> <td>0.02603</td> <td>529</td> <td>0.90</td> <td>0.3701</td> </tr> <tr> <td>WPstressLag1</td> <td>0.06751</td> <td>0.02418</td> <td>494</td> <td>2.79</td> <td>0.0054</td> </tr> </table>	Effect	Estimate	Error	DF	t Value	Pr > t	Intercept	1.9860	0.1274	120	15.59	<.0001	PMstress2	0.4723	0.1107	122	4.27	<.0001	WPstressLag0	0.02335	0.02603	529	0.90	0.3701	WPstressLag1	0.06751	0.02418	494	2.79	0.0054
Effect	Estimate	Error	DF	t Value	Pr > t																										
Intercept	1.9860	0.1274	120	15.59	<.0001																										
PMstress2	0.4723	0.1107	122	4.27	<.0001																										
WPstressLag0	0.02335	0.02603	529	0.90	0.3701																										
WPstressLag1	0.06751	0.02418	494	2.79	0.0054																										
<p>Random Intercept + 3-Lag Heterogeneous Toeplitz R:</p> <p>Fit Statistics</p> <p>-2 Res Log Likelihood 895.6</p> <p>AIC (smaller is better) 917.6</p> <p>AICC (smaller is better) 918.0</p> <p>BIC (smaller is better) 948.5</p>	<p>Solution for Fixed Effects</p> <p>Standard</p> <table border="0"> <tr> <th>Effect</th> <th>Estimate</th> <th>Error</th> <th>DF</th> <th>t Value</th> <th>Pr > t </th> </tr> <tr> <td>Intercept</td> <td>1.9649</td> <td>0.1307</td> <td>121</td> <td>15.04</td> <td><.0001</td> </tr> <tr> <td>PMstress2</td> <td>0.4833</td> <td>0.1132</td> <td>121</td> <td>4.27</td> <td><.0001</td> </tr> <tr> <td>WPstressLag0</td> <td>0.01976</td> <td>0.02662</td> <td>557</td> <td>0.74</td> <td>0.4582</td> </tr> <tr> <td>WPstressLag1</td> <td>0.05919</td> <td>0.02509</td> <td>545</td> <td>2.36</td> <td>0.0187</td> </tr> </table>	Effect	Estimate	Error	DF	t Value	Pr > t	Intercept	1.9649	0.1307	121	15.04	<.0001	PMstress2	0.4833	0.1132	121	4.27	<.0001	WPstressLag0	0.01976	0.02662	557	0.74	0.4582	WPstressLag1	0.05919	0.02509	545	2.36	0.0187
Effect	Estimate	Error	DF	t Value	Pr > t																										
Intercept	1.9649	0.1307	121	15.04	<.0001																										
PMstress2	0.4833	0.1132	121	4.27	<.0001																										
WPstressLag0	0.01976	0.02662	557	0.74	0.4582																										
WPstressLag1	0.05919	0.02509	545	2.36	0.0187																										
<p>AR(1) in R only:</p> <p>Fit Statistics</p> <p>-2 Res Log Likelihood 970.9</p> <p>AIC (smaller is better) 974.9</p> <p>AICC (smaller is better) 975.0</p> <p>BIC (smaller is better) 980.6</p>	<p>Solution for Fixed Effects</p> <p>Standard</p> <table border="0"> <tr> <th>Effect</th> <th>Estimate</th> <th>Error</th> <th>DF</th> <th>t Value</th> <th>Pr > t </th> </tr> <tr> <td>Intercept</td> <td>1.9868</td> <td>0.1218</td> <td>131</td> <td>16.31</td> <td><.0001</td> </tr> <tr> <td>PMstress2</td> <td>0.4828</td> <td>0.1055</td> <td>132</td> <td>4.57</td> <td><.0001</td> </tr> <tr> <td>WPstressLag0</td> <td>0.03198</td> <td>0.02866</td> <td>613</td> <td>1.12</td> <td>0.2650</td> </tr> <tr> <td>WPstressLag1</td> <td>0.05776</td> <td>0.02726</td> <td>614</td> <td>2.12</td> <td>0.0345</td> </tr> </table>	Effect	Estimate	Error	DF	t Value	Pr > t	Intercept	1.9868	0.1218	131	16.31	<.0001	PMstress2	0.4828	0.1055	132	4.57	<.0001	WPstressLag0	0.03198	0.02866	613	1.12	0.2650	WPstressLag1	0.05776	0.02726	614	2.12	0.0345
Effect	Estimate	Error	DF	t Value	Pr > t																										
Intercept	1.9868	0.1218	131	16.31	<.0001																										
PMstress2	0.4828	0.1055	132	4.57	<.0001																										
WPstressLag0	0.03198	0.02866	613	1.12	0.2650																										
WPstressLag1	0.05776	0.02726	614	2.12	0.0345																										
<p>Random Intercept Only (CS):</p> <p>Fit Statistics</p> <p>-2 Res Log Likelihood 1028.2</p> <p>AIC (smaller is better) 1032.2</p> <p>AICC (smaller is better) 1032.2</p> <p>BIC (smaller is better) 1037.8</p>	<p>Solution for Fixed Effects</p> <p>Standard</p> <table border="0"> <tr> <th>Effect</th> <th>Estimate</th> <th>Error</th> <th>DF</th> <th>t Value</th> <th>Pr > t </th> </tr> <tr> <td>Intercept</td> <td>1.9522</td> <td>0.1307</td> <td>122</td> <td>14.94</td> <td><.0001</td> </tr> <tr> <td>PMstress2</td> <td>0.4872</td> <td>0.1131</td> <td>122</td> <td>4.31</td> <td><.0001</td> </tr> <tr> <td>WPstressLag0</td> <td>0.04558</td> <td>0.02926</td> <td>633</td> <td>1.56</td> <td>0.1199</td> </tr> <tr> <td>WPstressLag1</td> <td>0.05712</td> <td>0.02831</td> <td>632</td> <td>2.02</td> <td>0.0440</td> </tr> </table>	Effect	Estimate	Error	DF	t Value	Pr > t	Intercept	1.9522	0.1307	122	14.94	<.0001	PMstress2	0.4872	0.1131	122	4.31	<.0001	WPstressLag0	0.04558	0.02926	633	1.56	0.1199	WPstressLag1	0.05712	0.02831	632	2.02	0.0440
Effect	Estimate	Error	DF	t Value	Pr > t																										
Intercept	1.9522	0.1307	122	14.94	<.0001																										
PMstress2	0.4872	0.1131	122	4.31	<.0001																										
WPstressLag0	0.04558	0.02926	633	1.56	0.1199																										
WPstressLag1	0.05712	0.02831	632	2.02	0.0440																										
<p>Residual Variance Only (VC):</p> <p>Fit Statistics</p> <p>-2 Res Log Likelihood 1883.4</p> <p>AIC (smaller is better) 1885.4</p> <p>AICC (smaller is better) 1885.4</p> <p>BIC (smaller is better) 1888.2</p>	<p>Solution for Fixed Effects</p> <p>Standard</p> <table border="0"> <tr> <th>Effect</th> <th>Estimate</th> <th>Error</th> <th>DF</th> <th>t Value</th> <th>Pr > t </th> </tr> <tr> <td>Intercept</td> <td>1.9259</td> <td>0.05754</td> <td>752</td> <td>33.47</td> <td><.0001</td> </tr> <tr> <td>BPstress2</td> <td>0.4842</td> <td>0.04999</td> <td>752</td> <td>9.69</td> <td><.0001</td> </tr> <tr> <td>WPstressLag0</td> <td>0.006840</td> <td>0.06633</td> <td>752</td> <td>0.10</td> <td>0.9179</td> </tr> <tr> <td>WPstressLag1</td> <td>0.06815</td> <td>0.06450</td> <td>752</td> <td>1.06</td> <td>0.2911</td> </tr> </table>	Effect	Estimate	Error	DF	t Value	Pr > t	Intercept	1.9259	0.05754	752	33.47	<.0001	BPstress2	0.4842	0.04999	752	9.69	<.0001	WPstressLag0	0.006840	0.06633	752	0.10	0.9179	WPstressLag1	0.06815	0.06450	752	1.06	0.2911
Effect	Estimate	Error	DF	t Value	Pr > t																										
Intercept	1.9259	0.05754	752	33.47	<.0001																										
BPstress2	0.4842	0.04999	752	9.69	<.0001																										
WPstressLag0	0.006840	0.06633	752	0.10	0.9179																										
WPstressLag1	0.06815	0.06450	752	1.06	0.2911																										